The two examples are based on Exercise 28. The result should a square.

```
> with(plots);
```

The contour integral below starts at 0 and goes to z along a straight path. Since we have path independence this path is as good as any other.

```
> fsquare := z-> z*int( (( (z*t-1)*(z*t-I)*(z*t+1)*(z*t+I) )^(-1/2) ), t=0..1);
```

$$fsquare := z \to z \left(\int_0^1 \frac{1}{\sqrt{(z\,t-1)\,(z\,t-\mathrm{I})\,(z\,t+1)\,(z\,t+\mathrm{I})}}\,\mathrm{d}t \right) \tag{1}$$

```
> complexplot(fsquare(cos(theta)+I*sin(theta)),theta=0..2*Pi,view=[-2..2,-2..2]);   # Image of unit circle. Should be a square.
```



If we restrict theta, we can start to see what is happening

```
> complexplot(fsquare(cos(theta)+I*sin(theta)),theta=0..Pi/4+0.1,
  view=[-2..2,-2..2]);
```



It is jumping across and turning back at Pi/4.

```
> s1:=complexplot( fsquare(cos(theta)+I*sin(theta)),theta=0..Pi/4,
  view=[-2..2,-2..2]):
> s2:=complexplot(-fsquare(cos(theta)+I*sin(theta)),theta=
  Pi/4+0.001..Pi/2-0.001,view=[-2..2,-2..2]):
> s3:=complexplot( fsquare(cos(theta)+I*sin(theta)),theta=
  Pi/2+0.001..3*Pi/4-0.001,view=[-2..2,-2..2]):
> s4:=complexplot(-fsquare(cos(theta)+I*sin(theta)),theta=3*
  Pi/4+0.001..Pi-0.001,view=[-2..2,-2..2]):
> s5:=complexplot( fsquare(cos(theta)+I*sin(theta)),theta=Pi+0.001.
  .5*Pi/4-0.001,view=[-2..2,-2..2]):
> s6:=complexplot(-fsquare(cos(theta)+I*sin(theta)),theta=5*
  Pi/4+0.001..6*Pi/4-0.001,view=[-2..2,-2..2]):
> s7:=complexplot( fsquare(cos(theta)+I*sin(theta)),theta=6*
  Pi/4+0.001..7*Pi/4-0.001,view=[-2..2,-2..2]):
```

```
> s8:=complexplot(-fsquare(cos(theta)+I*sin(theta)),theta=7*
  Pi/4+0.001..2*Pi-0.001,view=[-2..2,-2..2]):
> display(s1,s2,s3,s4,s5,s6,s7,s8);
```




```
> complexplot(fsquare(.9*cos(theta)+0.9*I*sin(theta)),theta=0..2*
  Pi,view=[-2..2,-2..2]);
```



Next example, also based on Exercise 28, but should give a pentagon.

```
> f2:= z-> z*(int( (( (z*t-1)*(z*t-I)*(z*t+1)*(z*t+I)*(z*
  t-0.70710678-I*0.70710678) )^(-0.4)), t=0..1,digits=3,numeric=
  true) ) ;
```

$$f2 := z \rightarrow z\left(int\left(\left((z\,t-1)\,(z\,t-\mathrm{I})\,(z\,t+1)\,(z\,t+\mathrm{I})\,(z\,t-0.70710678-\mathrm{I}\cdot 0.70710678)\right)^{-0.4},\ t=0..1,\ digits=3,\ numeric=true\right)\right) \quad (2)$$

```
> B:=complexplot(f2(cos(theta)+I*sin(theta)),theta=0..2*Pi,view=
  [-2..2,-2..2],numpoints=300,color=black,thickness=3):
> r1:=complexplot(f2(0.9*cos(theta)+0.9*I*sin(theta)),theta=0..2*
  Pi,view=[-2..2,-2..2],numpoints=30,color=red):
> r2:=complexplot(f2(0.7*cos(theta)+0.7*I*sin(theta)),theta=0..2*
  Pi,view=[-2..2,-2..2],numpoints=30,color=red):
> r3:=complexplot(f2(0.5*cos(theta)+0.5*I*sin(theta)),theta=0..2*
  Pi,view=[-2..2,-2..2],numpoints=30,color=red):
> display(B,r1,r2,r3);
```



This example, based on Exercise 29 in 5.2, do not go well! It should form a rectangle.

```
> f3 := z-> z*int(  ((z*t)*(z*t-1)*(z*t-2))^(-1/2), t=0..1,numeric=
  true, digits=3 );
```

$$f3 := z \rightarrow z\left(int\left(\frac{1}{\sqrt{z\,t\,(z\,t-1)\,(z\,t-2)}}, t=0..1, numeric = true, digits = 3 \right) \right) \quad (3)$$

```
> complexplot(f3(x),x=-3..3,view=[-10..10,-10..10],numpoints=20);
```




```
> f3 := z-> evalf(z*int( Re( (z*t)^(-0.5)*(z*t-1)^(-0.5)*(z*t-2)^
  (-0.5)), t=0..1)) + I*evalf(z*int( Im( (z*t)^(-0.5)*(z*t-1)^
  (-0.5)*(z*t-2)^(-0.5)), t=0..1));
```

$$f3 := z \to evalf\left(z\left(\int_0^1 \Re\left((z\,t)^{-0.5}\,(z\,t-1)^{-0.5}\,(z\,t-2)^{-0.5}\right)\,\mathrm{d}t\right)\right) + \mathrm{I}\,evalf\left(z\left(\int_0^1 \Im\left((z\,t)^{-0.5}\,(z\,t-1)^{-0.5}\,(z\,t-2)^{-0.5}\right)\,\mathrm{d}t\right)\right) \quad (4)$$

```
> complexplot(f3(x),x=-3..3,view=[-10..10,-10..10]);
```




```
> for n from -30 to 30 do P[n]:=f3(n/10) end do:
> P[0]:=0.0:P[7];
```

$$-1.516363286 + 0.\,I \quad (5)$$

```
> pointplot({seq([Re(P[m]),Im(P[m])],m=-30..30)},view=[-10..10,-10..10],color=red);
```



Finally got a rectangle!